
1 はじめに

1.1 常微分方程式って何だったっけ — 復習

常微分方程式というのは大雑把に言うと、「一つの実独立変数 t の未知関数 x = x(t) を求

めるための問題で、x とその導関数
dx

dt
,

d2x

dt2
, · · ·, dkx

dtk
についての方程式になっているもの」

のことです。(以下では
dx

dt
= x′,

d2x

dt2
= x′′,

d3x

dt3
= x(3), · · · のような書き方もします。)

例 1: x′(t) = f(t) (f は既知関数)

例 2: x′′(t) = −g (g は既知の定数)

例 3: x′(t) = ax(t) (a は既知の定数)

例 4: x′′(t) = −ω2x(t) (ω は既知の定数)

例 5: x′′(t) + 2γx(t) + ω2x(t) = 0 (γ, ω は既知の定数)

ここに例としてあげた方程式はいずれも割とポピュラーなものなのですが、見覚えがあるで
しょうか？どの場合もこれらの方程式だけでは解が一つに定まらず、何らかの条件を付け足す
ことによって初めて解が決定されます。その条件として、ある特定の t の値 t0 に対する x の
値 x(t0) や導関数の値を指定するというタイプのものがよくありますが、そういうものを初期
条件と呼びます (これは t が時刻を表す変数で、t0 を現象が始まる時刻のように解釈するから
でしょう)。
例えば、上の例 1, 3 に対して

x(0) = x0 (x0 は既知定数),

例 2, 4, 5 に対して
x(0) = x0, x′(0) = v0 (x0, v0 は既知定数)

のように与えられた条件が初期条件です。また、初期条件を添えて解が決定されるようにした
問題を、(常微分方程式の)初期値問題と言います。

1.2 これからの目標

微分方程式は、他の諸科学への応用のみならず1、数学それ自体にとっても非常に重要です2。
ところが困ったことに、微分方程式は大抵の場合に、良く知られている関数で解を表現するこ
とが出来ません。これは解が存在しないということではありません。解はほとんどいつでも存
在するけれども、それを簡単な演算 (不定積分を取る、四則演算、逆関数を取る、初等関数に
代入するなど)で求める — いわゆる求積法で解く — ことは、よほど特殊な問題でない限り
出来ない、ということです。
この困った状況をある程度解決するのが、解を数値的に求める方法です。この情報処理 II

では、いくつかの基本的な数値解法を学んで、実際に常微分方程式を解いてみます。これは計
算機による数値シミュレーション3の典型例と呼べるものですし、マスターしておくと役に立
ちます。

2 数値解法 (1)

2.1 問題の設定と数値解法の基本原理

常微分方程式としては正規形4のもののみを扱います。後で例で見るように、高階の方程式
も一階の方程式に帰着されますから、当面一階の方程式のみを考えます。独立変数を t、未知
関数を x = x(t) とすれば、一階正規形の常微分方程式とは

dx

dt
= f(t, x) (t ∈ [a, b])(1)

の形に表わされる方程式のことです。ここで f は既知の関数です。初期条件としては

x(a) = x0 (x0 は既知定数)(2)

の形のものを考えます。x0 は既知の定数です。(1),(2) を同時に満たす関数 x(t) を求めよ、と
いうのが一階正規形常微分方程式の初期値問題です。この時関数 x(t) を初期値問題 (1),(2) の
解と呼びます。

常微分方程式の数値解法の基本的な考え方は次のようなものです。「問題となっている区間
[a, b] を

a = t0 < t1 < t2 < · · · < tN = b

と分割して、各 “時刻”tj での x の値 xj = x(tj) (j = 1, 2, . . . , N) を近似的に求めることを目
標とする。そのために微分方程式 (1) から {xj}j=0,...,N を解とする適当な 差分方程式5を作り、
それを解く。」

1微分方程式は物理学の問題を扱うために発明されましたが、現在では自然科学以外でも応用されています。
2常微分方程式の簡単なものは高等学校でも学びましたし、1 年次にも微分方程式という授業がありました。

数学科の 3 年次にもより詳しいことを学ぶための講義があります。常微分方程式に対する参考書は色々あります
が、例えば、 3 年生向けの講義の教科書になっている、笠原晧司著「微分方程式の基礎」朝倉書店、をあげてお
きます。

3simulation(模擬実験)を「シュミレーション」と読み間違えないでください。「シミュレーション」ですから
ね。

4方程式が最高階の導関数について解かれている、ということですが、よく分からなくても差し支えありませ
ん。

5漸化式のようなものだと思って構いません。差分とは、高等学校の数列で言う階差のことです。

区間 [a, b] の分割の仕方ですが、以下では簡単のため N 等分することにします。つまり

h = (b− a)/N, tj = a + jh.

となります。

2.2 Euler(オイラー)法の紹介

微分 x′(t) =
dx

dt
は差分商

x(t + h)− x(t)

h
の h → 0 の極限です。そこで、(1) 式の微分を差

分商で置き換えて近似することによって、次の方程式を得ます。

xj+1 − xj

h
= f(tj, xj) (j = 0, 1, . . . , N − 1)

変形すると
xj+1 = xj + hf(tj, xj).(3)

これを漸化式として使って、x0 から順に x1, x2, . . ., xN が計算出来ます。この方法を Euler(オ
イラー)法と呼びます。
こうして得られる N + 1 個の点 (tj, xj) を順に結んで得られる折れ線関数は（f に関する適
当な仮定のもとで）N → +∞ の時（h = (b− a)/N について言えば h → 0）、真の解 x(t) に
収束することが証明できます6。ここでは簡単な例で収束を確かめてみましょう。

2.3 プログラミングの仕方

初期値 x0 が与えられたとき、漸化式 (3) によって、数列 {xj}j=1,···,N を計算するプログラ
ムはどう作ったらよいでしょうか？ここでは二つの素朴なやり方を紹介しましょう。

配列を使う方法 数列を配列で表現するのは、Fortran では自然な発想です。例えば

integer MAXN
parameter (MAXN = 1000)
real x(0:MAXN)

のように配列 “x” を用意しておいて

x(0) = x0
t = a
do j=0,N-1
x(j+1) = x(j) + h * f(t,x(j))
t = t + h

end do

とするわけです。

6現在の数学科のカリキュラムでは 3 年次に開講されている常微分方程式の講義で学びます。

配列を使わないですませる方法 漸化式 (3) を解くために、配列は絶対必要というわけではあ
りません7。例えば、変数 “x” に各段階の xj の値を収めておくとして

x = x0
t = a
do j = 0,N-1
x = x + h * f(t,x)
t = t + h

end do

のようなプログラムで計算が出来ます。

重箱の隅をつつく注意: “t = t + h” とすると、誤差が蓄積されがちです。これを避けるには、次
のように書き換えると良いでしょう。

t = a + (j + 1) * h

2.4 例題

例 1: 初期値問題
x′(t) = x(t) (t ∈ [0, 1]), x(0) = 1

の解は x(t) = et であるが、Euler 法を用いて解くと xN =

(
1 +

1

N

)N

となる。したがって、

確かに N → +∞ の時に xN → e = x(1) となっている。

例題 6.1 Euler 法を用いて、例 1 の初期値問題を解くプログラムを作って収束を調べよ。

* reidai6-1.f -- 微分方程式の初期値問題を Euler 法で解く
program rei61

* 開始時刻と終了時刻
real a,b
parameter (a=0.0,b=1.0)

* 変数と関数の宣言
integer N,j
real t,x,h,x0,f
external f

* 初期値
x0 = 1.0

* 区間の分割数 N を入力してもらう
write(*,*) ’ N=’
read(*,*) N

* 小区間の幅
h=(b-a)/N

* 開始時刻と初期値のセット
t=a
x=x0

7配列はメモリーを消費しますし、(特に Fortran の場合、配列の大きさは実行時に変更できないので)プログ
ラムを書く際に、どれくらいの大きさの配列を用意したらいいのかという問題に悩まなくてはなりません。

write(*,*) t,x
* Euler 法による計算

do j=0,N-1
x=x+h*f(t,x)
t=t+h
write(*,*) t,x

end do
end

**********************************************************************
* 微分方程式 x’=f(t,x) の右辺の関数 f の定義

real function f(t,x)
real t,x
f=x
end

このプログラムをコンパイルして実行すると、分割数 N を尋ねてきますので、色々な値を入
力して試してみて下さい。各時刻 tj における xj の値 (j = 0, 1, · · · , N) を画面に出力します。
確認用にいくつかの N の値に対する場合の、x(1) = xN の値を書いておきます。N = 10 の
場合 xN = 2.59374261, N = 100の場合 xN = 2.70481372, N = 1000の場合 xN = 2.71692038,

N = 10000 の場合 xN = 2.71814346, · · ·.
分割数 N が大きくなるほど、真の値 e = 2.7182818284590452 · · · に近付いていくはずです。
問題 6.1: 例題 6-1 とは異なる初期値問題を適当に設定して、それを Euler 法で解いてみよ。
(結果についてもきちんと吟味すること。)

問題 6.2: “reidai6-1.f” の出力するデータを用いて、解曲線 (関数 t 7→ x(t) のグラフのこ
と)を描きなさい。

2.5 Euler 法の収束の速さ

先ほどの実験では Euler 法による解は N が大きくなればなるほど真の解に近くなるはずで
す。実際 N → +∞ とすると真の解に収束することが証明できるわけなのですが、それでは、
どれくらいの速さで収束するのでしょうか？これを実験的に調べてみましょう。

例題 6.2: 例題 6.1 と同じ初期値問題で、色々な分割数 N に対して問題を時、t = 1 での誤差
の大きさ |x(1)− xN | = |e− xN | について調べよ。
こういう場合は、色々な N に対して一斉に |e− xN | を計算するプログラムを作るのがいい
でしょう。

* reidai6-2.f --- 常微分方程式の初期値問題を Euler 法で解く
program rei62

* 開始時刻と終了時刻
real a,b
parameter (a=0.0,b=1.0)

* 初期値
real x0

* 変数と関数の宣言
real t,x,h,f,e
integer N0,N1,N2,N,i

* 自然対数の底 e (=2.7182818284..)

e=exp(1.0)
* 初期値の設定

x0=1.0
* どういう N について計算するか？
* N0 から N1 まで、N2 刻みで増える N に

write(*,*) ’ FIRSTN,LASTN,STEPN=’
read(*,*) N0,N1,N2

* 計算の開始
do N=N0,N1,N2

h=(b-a)/N
* 開始時刻と初期値のセット

t=a
x=x0

* Euler 法による計算
do i=0,N-1

x=x+h*f(t,x)
t=t+h

end do
write(*,*) N,abs(e-x)

end do
end

**********************************************************************
* 微分方程式 x’=f(t,x) の右辺の関数 f の定義 -- 前と同じだから略

コンパイルして実行すると “ FIRSTN,LASTN,STEPN=” と尋ねてきます。例えば “100 1000
50” と答えると 100 から 1000 まで 50 刻みで増やして行った値 100, 150, 200, · · ·, 900, 950,
1000を N として計算して、最終的に得られた誤差を出力します。実行結果を見てみましょう。

waltz11% reidai6-2
FIRSTN,LASTN,STEPN=

100 1000 50
100 1.34680271e-02
150 9.00602341e-03
200 6.76608086e-03
.... (途中略) ...
900 1.50966644e-03
950 1.42812729e-03
1000 1.36137009e-03

waltz11%

“mgraph”コマンドには、対数目盛によるグラフを描く機能 (起動時に -lx, -ly を指定しま
す。ここで “l” は “logarithmic”(=「対数の」) の頭文字です）があります。この場合は、例
えば次のようにしたりすると良いでしょう。

waltz11% reidai6-2 | mgraph -lx -ly | xplot

10,10000,100

このグラフから、誤差 = O(N−1) (N → +∞) であることが読みとれます。実はこれは
Euler 法の持つ一般的な性質です。
実は Euler 法は収束があまり速くないので、実際には特殊な場合を除いて使われていませ
ん。そこで、、、、

3 Runge–Kutta (ルンゲ–クッタ)法
前節で解説した Euler 法は簡単で、これですべてが片付けば喜ばしいのですが、残念なが
らあまり効率が良くありません。高精度の解を計算するためには、分割数 N をかなり大きく
取る (=大量の計算をする)必要があります。特別な場合8を除けば、実際に使われることは滅
多にないでしょう。率直にいって実用性は低いです。
より高精度の公式は、現在まで様々なものが開発されていますが、比較的簡単で、精度がま
あまあ高いものに Runge–Kutta 法と呼ばれるものがあります。それは xj から xj+1 を求める
漸化式として次のものを用います。

k1 = hf(tj, xj)

k2 = hf(tj + h/2, xj + k1/2)

k3 = hf(tj + h/2, xj + k2/2)

k4 = hf(tj + h, xj + k3)

xj+1 = xj +
1

6
(k1 + 2k2 + 2k3 + k4)

これがどうやって導かれたものかは解説しません。まずは使ってみましょう。

* reidai6-3.f -- 微分方程式の初期値問題を Runge-Kutta 法で解く
program rei63

8例えば微分方程式の右辺に現れる関数 f が、解析的な式で定義された滑らかなものではなく、実験により計
測されたデータにより定義されている場合等は、高精度の公式を用いるよりも、Euler 法の方が良いことがあり
ます。

* 開始時刻と終了時刻
real a,b
parameter (a=0.0,b=1.0)

* 変数と関数の宣言
integer N,j
real t,x,h,x0,f,k1,k2,k3,k4
external f

* 初期値
x0 = 1.0

* 区間の分割数 N を入力してもらう
write(*,*) ’ N=’
read(*,*) N

* 小区間の幅
h=(b-a)/N

* 開始時刻と初期値のセット
t=a
x=x0
write(*,*) t,x

* Runge-Kutta 法による計算
do j=0,N-1

k1 = h * f(t, x)
k2 = h * f(t + h / 2, x + k1 / 2)
k3 = h * f(t + h / 2, x + k2 / 2)
k4 = h * f(t + h, x + k3)
x = x + (k1 + 2 * k2 + 2 * k3 + k4) / 6
t = t + h
write(*,*) t,x

end do
end

**********************************************************************
* 微分方程式 x’=f(t,x) の右辺の関数 f の定義 --- これも前と同じだから略

コンパイル・実行の結果は次のようになるはずです。

waltz11% f77o reidai6-3.f
Fortran 90 diagnostic messages: program name(f)
jwd2008i-i ‘‘reidai6-3.f’’, line 30: この仮引数は,副プログラム中で使用されてい

ま
せん.(名前:t)
waltz11% reidai6-3
N=

10
0.000000000e+00 1.00000000
0.100000001 1.10517085
0.200000003 1.22140265
0.300000012 1.34985852
0.400000006 1.49182427
0.500000000 1.64872062
0.600000024 1.82211792
0.700000048 2.01375151
0.800000072 2.22553945
0.900000095 2.45960140

1.00000012 2.71827984
waltz11%

たった 10等分なのに相対誤差が 10−6以下になっています (∵ x(1) = e1 = e = 2.7182818284 · · ·
であるので、相対誤差=|e−2.71827984|/e = 7.31 · · ·×10−7)。Runge–Kutta法の公式は Euler

法よりは大部面倒ですが、それに見合うだけの価値があることが納得できるでしょう。
問題 6-3: “reidai6-3.f” を普通の “real”(単精度実数型)の代わりに “real*8” (倍精度実数
型)を使うように書き換えて、大きな N に対してどうなるか、実験しなさい。倍精度化のた
めの書き換えは、具体的には、(i) プログラムの中に何箇所かある “real” を “real*8” に書
き換える、(ii) 実数の定数を “0.0” → “0.0d0” のように、末尾に “d0” を付けたりして9、倍
精度実数定数に書き換える、 (iii) “real” や “complex” への型変換を指定しているとことが
あれば、それぞれ “real*8”, “complex*16” への型変換に書き換える、というようなことをや
ればいいが、今の場合は (i) だけでも十分動作するプログラムになる。
問題 6-4: 区間の分割数 N を変えながら

x′(t) = x (t ∈ [0, 1]), x(0) = 1

を Runge–Kutta 法を用いて解くプログラムを作り、t = 1 での誤差 |xN − x(1)| を調べよ (前
節 “reidai6-2.f” の真似をする)。Euler 法と比べてどうなるか？

4 対数グラフに関するメモ
中学・高校で、データをグラフにプロットして、正比例の関係にあることを確かめ、さらに
は比例定数を求める、ということをした経験があるでしょう。ここでは y = cxα のような関数
について、同じことをするにはどうしたら良いか、説明します。これには両辺の対数を取って、

log y = log(cxα) = log c + log xα = log c + α log x.

これから、log y = Y , log x = X, log c = C とおくと

Y = αX + C.

そこで、データ (xj, yj) に対し、(log xj, log yj) を座標に持つ点をプロットすると、点は直線
上に並び、傾きは α になるはずです。—— これが対数グラフの原理です。対数グラフは簡単
ですが、便利で役立つものです。
問題 6-5: y = ax のようなものは、どう調べたらいいか考え、実験して確かめよ。

9“d整数”は “e整数” のような指数形式の実数表現の指数部を表します。よって “1.0e-7” のような定数は
“1.0d-7” のように書き換えます。

